
1. [bookmark: _Toc122434485][bookmark: _Toc110976093]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.3.2 which is part of the LTE_A_PRO test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc110976094]Verification Test Summary 
Test Case: 	11.3.2
ATS Version:	iwd-TTCN3-B2020-09_D22wk23	
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
[bookmark: _Hlk109419132]UE used:	Qualcomm SA515M
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc110976095]Corrections required
1. [bookmark: _Toc110976096]Change 1
	Function name
	f_TC_11_3_2_EUTRA()

	Reason for change
	UE is already detached at the end of the test body

	Summary of change
	Corrected postamble state

	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change
  function f_TC_11_3_2_EUTRA() runs on EUTRA_PTC
{ /* eCall Only mode / T3445 / eCall inactivity procedure / Removal of eCall only restriction after a call to URI for test service */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2");
    
    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_2_Body();
    
    // Postamble
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E1_IDLE); // optional IMS Deregistration is handled in the default, before the UE detaches
  }    

After change
  function f_TC_11_3_2_EUTRA() runs on EUTRA_PTC
{ /* eCall Only mode / T3445 / eCall inactivity procedure / Removal of eCall only restriction after a call to URI for test service */

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c1, -, HANDLE_UM_DATA, USE_BIG_GRANTS);
    
    // Create cell and send out system information
    // Specific SIB1 support for IMS emergency support and eCall support will be set in default functions
    f_EUTRA_CellConfig_Def(eutra_CellA);
    
    // Insert USIM for eCall only enabled
    f_UT_USIM_Insert (UT, "36.508 Table 4.9.3.5-2");
    
    // Switch cell on
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);
    
    // The UE is in state Switched OFF (state 1)
    f_EUTRA_TestBody_Set(true);
    
    fl_TC_11_3_2_Body();
    
    // Postamble
    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E4_DEREGISTERED); // optional IMS Deregistration is handled in the default, before the UE detaches
  }  
    

1. [bookmark: _Toc110976097]Change 2
	Function name
	f_TC_11_3_2_Body()

	Reason for change
	1. Currently no handling for RRCConnectionRequest at step 4

1. Currently the SS is releasing the connection after the completion of IMS registration, during function f_EUTRA_InitialRegistration_Step16_18b1(). But since the INVITE message will come immediately after following IMS registration (possibly interleaved, see change 3),  SS should maintain the RRC connection.

1. Following change 2, SS is not releasing the connection so steps to handle RRCConnectionRequest and move back to CONNECTED state are not needed


1. Extra trigger to IMS PTC not needed, just need to wait for trigger from IMS to perform dedicated EPS bearer activation

1. Timer t_T3445Min timer is currently not  started. 

1. t_T3412 should be (re-)started following RRCConnectionRelease at step 67 according to note 2

1. T3412 value is format TV in TRACKING AREA UPDATE ACCEPT but currently template cs_GprsTimer_v is being used which does not include the IEI field.

1. Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach immediately.

1. DETACH REQUEST at step 73 will be integrity protected but not ciphered


	Summary of change
	1. RRC Connection request handled with emergency establishment cause

1. Replace the call to f_EUTRA_InitialRegistration_Step16_18b1() with a direct call to f_EUTRA_InitialRegistration_Step16_16c1(), so that the RRC connection is not released following the signalling. Also pass “dontWaitForIMS” to this function to avoid the 10s timer which can delay further test steps.

1. Removed the handling of RRC Connection request for 'mo-data' and service request procedure
 
1. Remove extra trigger to IMS PTC

1. Add starting of t_T3445Min

1. T3412 timer started after step 67.

1. Use template cs_GprsTimer_tv for T3412 ins TRACKING AREA UPDATE ACCEPT message

1. Pass NW feature support into f_EUTRA_TrackingAreaUpdate() indicating support of emergency bearer services, see change 2.4 

1. Correct security status for steps 72 and 73


Note : A prose CR on 36.523-1 associated with changes 2, 3 and 8 will be raised at RAN5#96.

	TTCN module
	 IMS_ECall.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change
  function fl_TC_11_3_2_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var ProcedureTransactionIdentifier v_EPS_TI  := '01'O;
    var APNandPCOs_Type v_APNandPCOs;
    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;
    var template (omit) ProtocolConfigOptions v_PcoToUE := omit;
    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(firstPDN);
    var template (value) GPRS_Timer v_T3412Value := cs_GprsTimer_v('010'B, '11111'B); //T3412 value = 186 (6 x 31) mins
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_AttDetValue_Type v_DetachType;
    var NAS_MSG_Indication_Type v_NasInd;
    var boolean v_T3445Min_elapsed := false;
    var float v_EcallDeregisteredInactive := 60.0;
    timer t_T3412Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 11160.0); // 186 minutes
    timer t_T3445Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 43200.0); // 12 hours
    timer t_T3445Min := f_EUTRA_SetTimerToleranceMin(eutra_CellA, nasTimer, 43200.0); // 12 hours

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(cs_EPS_NwkFtSup_IMSVoice_EMCBS); // @sic R5s180648 sic@ emergency bearer services in S1 mode and IMS voice over PS session in S1 mode are supported

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated.
    f_UT_InitiateECall(UT, TEST);

   
      
    //@siclog "Step 4" siclog@
    // Check: Does the UE sends an ATTACH REQUEST message with a PDN CONNECTIVITY REQUEST message to request PDN connectivity to the default PDN. EPS attach type = "combined EPS/IMSI attach"?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                         cr_508_ATTACH_REQUEST(tsc_EAttachType_Combined,
                                                                                                               v_AdditionalUpdateType),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                         f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);
    f_EUTRA_Security_Set ( v_SecurityParams );//Saving security parameters
   
    //@siclog "Step 5-13" siclog@
    // Steps 5-13 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_APNandPCOs := f_EUTRA_InitialRegistration_Step5_13(eutra_CellA, STATE2_IDLEUPDATE, v_NasInd);

    v_PcoFromUE := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;
    if (isvalue(v_APNandPCOs.protocolConfigurationOptions)) {
      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;
    }
    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE, v_PdnIndex);


    //@siclog "Step 14-15" siclog@
    // The SS transmits an ATTACH ACCEPT message with the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message and SS assigns value of 186 minutes for the timer T3412.
    f_EUTRA_InitialRegistration_Step14_15(eutra_CellA,
                                          NORMAL,
                                          v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType,
                                          v_EPS_TI,
                                          f_CheckPCOforIPallocationViaNas (v_PcoFromUE),
                                          v_PcoToUE,
                                          v_APNandPCOs.accessPointName,
                                          v_T3412Value);// 186 minutes

    //@siclog "Step 16-18b1" siclog@
    // Steps 16-18b1 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.

    f_EUTRA_InitialRegistration_Step16_18b1(eutra_CellA);

    //@siclog "Step 19" siclog@
    // Check: Does UE transmits a RRCConnectionRequest message with ' establishmentCause' set to 'mo-Data' '?
    //@siclog "Step 20-31" siclog@
    // The UE performs Steps 3-14 of the generic test procedure as described in TS 36.508 sub clause 4.5A.6.3
    // Send trigger to IMS PTC: Steps 9 to 11 are executed in IMS PTC
    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(eutra_CellA, -, -, -, -, -, -, -, -, false, cr_EstablishmentCause_MoData);    //Steps 3-8
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA); //Step 12-14
    //Wait for completion of IMS signalling
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
            
    //@siclog "Step 32-36" siclog@
    // Release eCall to URI for test service using the generic procedure described in TS 34.229-1 [35] sub clause C.33. @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // release the call
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
 
    //@siclog "Step 37" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
    // start after the end of the eCall
    t_T3412Max.start;
    t_T3445Max.start;
    
    //@siclog "Step 38" siclog@
    // The SS waits 1 minute.
    f_Delay(60.0);

    //@siclog "Step 39-62" siclog@
    // Steps 1 to 24 of the generic test procedure for IMS MT speech call specified in TS 36.508 subclause 4.5A.7.3
    // from function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
    f_EUTRA_RbEst_Def(eutra_CellA); //Steps 1-8
    // Send trigger to IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // Steps 9 to 12 are executed in IMS PTC
    //Wait Trigger from IMS PTC: steps 1-4 have been done
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA, -, -, -, true);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 62");

    //@siclog "Step 63-66" siclog@
    // Generic test procedure for MT release of IMS call as described in annex C.33 of TS 34.229-1 [35] takes place.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    //Wait Trigger from IMS PTC: MT release has happened
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
    
    //@siclog "Step 67" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    alt {
      //@siclog "Step 68" siclog@
      // Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message at the expiry of T3412 periodically for every 186 minutes and remains registered until T3445 expires?
      //@siclog "Step 69" siclog@
      // The SS transmits a TRACKING AREA UPDATE ACCEPT message.
      //@siclog "Step 70" siclog@
      // The UE sends TRACKING AREA UPDATE COMPLETE message.
      //@siclog "Step 71" siclog@
      // The SS releases the RRC connection.
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_TrackingAreaUpdate(eutra_CellA, tsc_RRC_TI_Def, -, cr_EstablishmentCause_Any, tsc_EpsUpdate_Periodic, -, -, v_T3412Value);
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 68");
          t_T3412Max.start;
          repeat;
        }
      [] t_T3412Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68 - TAU expected was not received in time");
        }
      [] t_T3445Min.timeout {
          v_T3445Min_elapsed := true;
          t_T3412Max.stop;
        }
      }

    v_DetachType :=  f_GetEDetachType(NORMAL);
    alt {
      //@siclog "Step 72" siclog@
      // Check: Does the UE transmit a DETACH REQUEST message at expiry of T3444?
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                     tsc_RRC_TI_Def,
                                                     ?,
                                                     (cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,
                                                                        cr_508_DETACH_REQUEST_MO(?,
                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                                 cr_DetachType( '0'B, v_DetachType)))));
          t_T3445Max.stop;
          if  (v_T3445Min_elapsed) {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 72");
          } else  {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was received too early");
          }
          //@siclog "Step 73" siclog@
          // The SS responds with DETACH ACCEPT message.
          SRB.send (cas_SRB1_NasPdu_REQ (eutra_CellA,
                                         cs_TimingInfo_Now,
                                         cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                        cs_508_DETACH_ACCEPT)));
        }
      [] t_T3445Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was not received");
        }
      }

    //@siclog "Step 74" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
  }   

After change
  function fl_TC_11_3_2_Body() runs on EUTRA_PTC
  {
    var GutiParameters_Type v_Guti1Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var ProcedureTransactionIdentifier v_EPS_TI  := '01'O;
    var APNandPCOs_Type v_APNandPCOs;
    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;
    var template (omit) ProtocolConfigOptions v_PcoToUE := omit;
    var PDN_Index_Type v_PdnIndex := f_EUTRA_InitialRegistrationPdnIndex(firstPDN);
    var template (value) GPRS_Timer v_T3412Value := cs_GprsTimer_tv('5A'O, '010'B, '11111'B); //T3412 value = 186 (6 x 31) mins
    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();
    var NAS_AttDetValue_Type v_DetachType;
    var NAS_MSG_Indication_Type v_NasInd;
    var boolean v_T3445Min_elapsed := false;
    var float v_EcallDeregisteredInactive := 60.0;
    timer t_T3412Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 11160.0); // 186 minutes
    timer t_T3445Max := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 43200.0); // 12 hours
    timer t_T3445Min := f_EUTRA_SetTimerToleranceMin(eutra_CellA, nasTimer, 43200.0); // 12 hours

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(cs_EPS_NwkFtSup_IMSVoice_EMCBS); // @sic R5s180648 sic@ emergency bearer services in S1 mode and IMS voice over PS session in S1 mode are supported

    //@siclog "Step 1" siclog@
    // The UE is switched on.
    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA);

    //@siclog "Step 2" siclog@
    // Wait for 60s for the UE to enter EMM-DEREGISTERED.eCALL-INACTIVE state
    f_Delay(v_EcallDeregisteredInactive);

    //@siclog "Step 3" siclog@
    // An eCall to URI for test service is initiated.
    f_UT_InitiateECall(UT, TEST);

	 f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_emergency_Access);
      
    //@siclog "Step 4" siclog@
    // Check: Does the UE sends an ATTACH REQUEST message with a PDN CONNECTIVITY REQUEST message to request PDN connectivity to the default PDN. EPS attach type = "combined EPS/IMSI attach"?
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                         cr_508_ATTACH_REQUEST(tsc_EAttachType_Combined,
                                                                                                               v_AdditionalUpdateType),
                                                                                         cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                                                                                                         f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    
    v_EPS_TI := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_NB_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);
    f_EUTRA_Security_Set ( v_SecurityParams );//Saving security parameters
   
    //@siclog "Step 5-13" siclog@
    // Steps 5-13 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    v_APNandPCOs := f_EUTRA_InitialRegistration_Step5_13(eutra_CellA, STATE2_IDLEUPDATE, v_NasInd);

    v_PcoFromUE := v_NasInd.Pdu.PiggybackedPduList[0].Msg.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;
    if (isvalue(v_APNandPCOs.protocolConfigurationOptions)) {
      v_PcoFromUE := v_APNandPCOs.protocolConfigurationOptions;
    }
    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE, v_PdnIndex);


    //@siclog "Step 14-15" siclog@
    // The SS transmits an ATTACH ACCEPT message with the ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message and SS assigns value of 186 minutes for the timer T3412.
    f_EUTRA_InitialRegistration_Step14_15(eutra_CellA,
                                          NORMAL,
                                          v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType,
                                          v_EPS_TI,
                                          f_CheckPCOforIPallocationViaNas (v_PcoFromUE),
                                          v_PcoToUE,
                                          v_APNandPCOs.accessPointName,
                                          v_T3412Value);// 186 minutes

    //@siclog "Step 16-18b1" siclog@
    // Steps 16-18b1 of the generic procedure for UE registration specified in TS 36.508 subclause 4.5.2.3 are performed.
    f_EUTRA_InitialRegistration_Step16_16c1 (eutra_CellA, dontWaitForIMS);

    //@siclog "Step 19" siclog@
    // Check: Does UE transmits a RRCConnectionRequest message with ' establishmentCause' set to 'mo-Data' '?
    //@siclog "Step 20-31" siclog@
    // The UE performs Steps 3-14 of the generic test procedure as described in TS 36.508 sub clause 4.5A.6.3
    // Send trigger to IMS PTC: Steps 9 to 11 are executed in IMS PTC
    //REMOVED f_EUTRA_ServiceRequestAndActivate_SRB2_DRB(eutra_CellA, -, -, -, -, -, -, -, -, false, cr_EstablishmentCause_MoData);    //Steps 3-8
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 19");
    //REMOVED f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA); //Step 12-14
    //Wait for completion of IMS signalling
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]); 
            
    //@siclog "Step 32-36" siclog@
    // Release eCall to URI for test service using the generic procedure described in TS 34.229-1 [35] sub clause C.33. @sic R5-196019 sic@
    //IMS call has been released.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); 
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
 
    //@siclog "Step 37" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
 t_T3445Min.start;
    t_T3445Max.start;
    //MOVED t_T3412Max.start;

    
    //@siclog "Step 38" siclog@
    // The SS waits 1 minute.
    f_Delay(60.0);

    //@siclog "Step 39-62" siclog@
    // Steps 1 to 24 of the generic test procedure for IMS MT speech call specified in TS 36.508 subclause 4.5A.7.3
    // from function f_EUTRA_GenericIMS_MT_SpeechCall(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC
    f_EUTRA_RbEst_Def(eutra_CellA); //Steps 1-8
    // Send trigger to IMS PTC
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]); // Steps 9 to 12 are executed in IMS PTC
    //Wait Trigger from IMS PTC: steps 1-4 have been done
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);

    f_EUTRA_ActivateDedicatedEpsBearer_Speech(eutra_CellA, -, -, -, true);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 62");

    //@siclog "Step 63-66" siclog@
    // Generic test procedure for MT release of IMS call as described in annex C.33 of TS 34.229-1 [35] takes place.
    f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_IMS1]);
    
    //Wait Trigger from IMS PTC: MT release has happened
    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_IMS1]);
    // The SS deactivates the EPS bearer. [This is the rest of C.33]
    f_EUTRA_DeactivateEPS_BearerContext (eutra_CellA,
                                         tsc_EpsDedicatedBearerId);
    
    //@siclog "Step 67" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    t_T3412Max.start;
 
    alt {
      //@siclog "Step 68" siclog@
      // Check: Does the UE transmit a TRACKING AREA UPDATE REQUEST message at the expiry of T3412 periodically for every 186 minutes and remains registered until T3445 expires?
      //@siclog "Step 69" siclog@
      // The SS transmits a TRACKING AREA UPDATE ACCEPT message.
      //@siclog "Step 70" siclog@
      // The UE sends TRACKING AREA UPDATE COMPLETE message.
      //@siclog "Step 71" siclog@
      // The SS releases the RRC connection.
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          f_EUTRA_TrackingAreaUpdate(eutra_CellA, tsc_RRC_TI_Def, -, cr_EstablishmentCause_Any, tsc_EpsUpdate_Periodic, -, -, v_T3412Value, cs_EPS_NwkFtSup_IMSVoice_EMCBS);
          f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 68");
          t_T3412Max.start;
          repeat;
        }
      [] t_T3412Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 68 - TAU expected was not received in time");
        }
      [] t_T3445Min.timeout {
          v_T3445Min_elapsed := true;
          t_T3412Max.stop;
        }
      }

    v_DetachType :=  f_GetEDetachType(NORMAL);
    alt {
      //@siclog "Step 72" siclog@
      // Check: Does the UE transmit a DETACH REQUEST message at expiry of T3444?
      [] SRB.check(receive(car_SRB0_RrcPdu_IND(?, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)))) {
          v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,
                                                     tsc_RRC_TI_Def,
                                                     ?,
                                                     (cr_NAS_Indication((tsc_SHT_IntegrityProtected),
                                                                        cr_508_DETACH_REQUEST_MO(?,
                                                                                                 f_GutiParameters2MobileIdentity(omit, v_Guti1Params),
                                                                                                 cr_DetachType( '0'B, v_DetachType)))));
          t_T3445Max.stop;
          if  (v_T3445Min_elapsed) {
            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 72");
          } else  {
            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was received too early");
          }
          //@siclog "Step 73" siclog@
          // The SS responds with DETACH ACCEPT message.
          SRB.send (cas_SRB1_NasPdu_REQ (eutra_CellA,
                                         cs_TimingInfo_Now,
                                         cs_NAS_Request(tsc_SHT_IntegrityProtected,
                                                        cs_508_DETACH_ACCEPT)));
        }
      [] t_T3445Max.timeout {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 72 - DETACH REQUEST expected was not received");
        }
      }

    //@siclog "Step 74" siclog@
    // The SS releases the RRC connection.
    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);
    
  }       
    

1. [bookmark: _Toc110976098]Change 3
	Function name
	f_TC_11_3_2_IMS1()

	Reason for change
	1. Following the completiong of IMS registration (receiving 200 OK for REGISTER), the UE may initiate the test eCall immediately, in parallel with subnsribing to the reg-event package. In particular the INVITE may be received at the TTCN before the 200 OK response for NOTIFY.

Note : This seems to be the only scenario  (eCall to test number) where the UE will proceed with call setup immediately following registration, and also subscribe to the reg-event pacakge

2. Remaining signalling for MO call setup missing

3. Wait for trigger from EUTRA not needed, see change 2.4

4. Missing steps for MT call setup

	Summary of change
	1.  Move some of the functionality of f_IMS_Registration() into this function. After sending of 200 OK for SUBSCRIBE and then NOTIFY, activate a default handler to receive the 200 OK for NOTIFY message in parallel with eCall setup steps, using new altstep (see change 9).

2. Add functions for C.21 steps 5-13, including new function f_IMS_eCallSetup_Send200OK_ReceiveACK_Step12_13(), see change 8.

3. Remove wait for trigger

4. Add remaining steps for C.11 MT call setup sequence

	TTCN module
	 IMS_ECall_IMS.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change


  function f_TC_11_3_2_IMS1() runs on IMS_PTC
  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var SipUrl v_ContactUrl;
    var INVITE_Request v_InviteRequest;

    f_IMS_PTC_Init(PDN_1);
    f_IMS_Registration();

    v_InviteRequestWithSdp := f_ECall_InitialInvite_AnnexC47(TEST); /* @sic R5-220971, R5-221073 sic@ */

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp); // steps 3-4
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl); // @sic R5-196019 sic@
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequestWithSdp := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdp);
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl);
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistration();
  }
 

After change
  
  import from IMS_ASP_TypeDefs all; 
  import from IMS_ASP_Templates all;
  import from IMS_MessageBody_Templates all;
  import from IMS_Procedures_Common all;
  import from IP_ASP_TypeDefs all;
  import from CommonDefs all;
  import from IMS_CommonFunctions all;
  
…………………………………………………


function f_TC_11_3_2_IMS1() runs on IMS_PTC
  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var SipUrl v_ContactUrl;
    var INVITE_Request v_InviteRequest;
    var MessageHeader v_MessageHeader_Response183; 
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdpVal; 
    var IMS_ResponseWithSdp_Type v_ResponseWithSdp; 
    var IMS_Response_Type v_SessionInProgress183;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var template (value) MessageHeader v_MessageHeader_Notify;
    var REGISTER_Request v_RegisterRequest;
    var InternetProtocol_Type v_Protocol;
    var default v_WaitFor200OK_NotifyEvent_Handler := null;

    f_IMS_PTC_Init(PDN_1);

    v_IMS_DATA_REQ := f_IMS_Register();
    f_IMS_Register_Subscribe();
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_RegisterRequest.msgHeader);
    v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    v_MessageHeader_Notify := f_IMS_NotifyRequest_MessageHeader_A_1_6(cs_SubscriptionState_Active); 
    IMS_Client.send(cas_IMS_Notify_Request(cs_IMS_RoutingInfo(f_IMS_PTC_ImsInfo_GetTransportProtocol()),    cs_NOTIFY_Request(v_ContactUrl, v_MessageHeader_Notify, cs_MessageBody_RegInfo(f_IMS_XmlMessageForNotify_Active(v_RegisterRequest)))));

    v_WaitFor200OK_NotifyEvent_Handler := activate(a_IMS_ReceiveResponse200OK_repeat(v_MessageHeader_Notify, v_Protocol))

    v_InviteRequestWithSdp := f_ECall_InitialInvite_AnnexC47(TEST); /* @sic R5-220971, R5-221073 sic@ */

    //REMOVED f_IMS_IPCAN_WaitForTrigger(IPCAN); 
    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp); // steps 3-4
    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_MOCallSetup_AnnexC21_Steps5_8(v_InviteRequestWithSdp, v_MessageHeader_Response183); 
    f_IMS_MOCallSetup_AnnexC21C25_Step9_11(v_InviteRequestWithSdp); 
    f_IMS_eCallSetup_Send200OK_ReceiveACK_Step12_13 (v_InviteRequestWithSdp); 
   
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);
    f_IMS_CallReleaseMT(v_ContactUrl); // @sic R5-196019 sic@
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    deactivate(v_WaitFor200OK_NotifyEvent_Handler);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);

    v_InviteRequestWithSdpVal := f_IMS_MTCallSetup_BuildInviteRequestWithSdp();
    v_InviteRequest := v_InviteRequestWithSdpVal.Invite; 
    v_ResponseWithSdp := f_IMS_MTCallSetup_AnnexC11_Steps1_4(v_InviteRequestWithSdpVal); 
    v_SessionInProgress183 := v_ResponseWithSdp.Response; 
    f_IMS_MTCallSetup_SendPRACK_ReceiveOK(v_InviteRequest, v_SessionInProgress183.msgHeader); 

    // Inform EUTRA: IMS Call established.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IPCAN_WaitForTrigger(IPCAN);
 
    //MOVED v_InviteRequest := v_InviteRequestWithSdpVal.Invite;

    v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader); 
    f_IMS_MTCallSetup_AnnexC11_Steps7_13(v_InviteRequestWithSdpVal, v_ResponseWithSdp);   
  
    f_IMS_CallReleaseMT(v_ContactUrl);
    // Inform EUTRA: IMS Call released.
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    f_IMS_IpcanReleaseWithOptionalDeregistration();
  }


[bookmark: _Toc110976099]2.4 Change 4
	Function name
	f_EUTRA_TrackingAreaUpdate()

	Reason for change
	Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach.


	Summary of change
	New parameter for specific NW feature support settings

	TTCN module
	 EUTRA_NASSteps.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change

function f_EUTRA_TrackingAreaUpdate(EUTRA_CellId_Type p_CellId,
                                      RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                      EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                      template (present) EstablishmentCause p_EstablishmentCause := ?, //@sic R5s100539 sic@
                                      template B3_Type p_EpsUpdate_TypeValue := omit,    // @sic R5s120617 sic@
                                      UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,
                                      B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B))// @sic R5-165962 sic@
    runs on EUTRA_PTC
  { /* @sic R5s100707 additional MCC160 comments sic@ */
    
    /* Step 1..5: */
    f_EUTRA_TrackingAreaUpdate_Step1_5(p_CellId, p_RRC_TI, p_ForcedAttach, p_EstablishmentCause, -, p_EpsUpdate_TypeValue, p_UE_NetworkCap, p_ActiveFlag, p_T3412); // @sic R5-160777, R5-180728 sic@
    
    /* Step 6: Release RRC connection */
    f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

  }

After change
---- Skipped code ---- 
 
function f_EUTRA_TrackingAreaUpdate(EUTRA_CellId_Type p_CellId,
                                      RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                      EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                      template (present) EstablishmentCause p_EstablishmentCause := ?, //@sic R5s100539 sic@
                                      template B3_Type p_EpsUpdate_TypeValue := omit,    // @sic R5s120617 sic@
                                      UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,
                                      B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                      template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B),
                                      template (value) EPS_NetworkFeatureSupport p_NtwFeatureSupport := cs_EPS_NwkFtSup_IMSVoice ) // @sic R5-165962 sic@
    runs on EUTRA_PTC
  { /* @sic R5s100707 additional MCC160 comments sic@ */
    
    /* Step 1..5: */
    f_EUTRA_TrackingAreaUpdate_Step1_5(p_CellId, p_RRC_TI, p_ForcedAttach, p_EstablishmentCause, -, p_EpsUpdate_TypeValue, p_UE_NetworkCap, p_ActiveFlag, p_T3412, p_NtwFeatureSupport); // @sic R5-160777, R5-180728 sic@
    
    /* Step 6: Release RRC connection */
    f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);
  }

[bookmark: _Toc110976100]2.5 Change 5
	Function name
	f_EUTRA_TrackingAreaUpdate_Step1_5()

	Reason for change
	Currently the TRACKING AREA UPDATE ACCEPT message sent from function f_EUTRA_TrackingAreaUpdate() does not indicate support of emergency bearer services, this can cause an eCall-Only device to detach.


	Summary of change
	New parameter for specific NW feature support settings, pass to f_EUTRA_TrackingAreaUpdate_Accept_Complete()

	TTCN module
	 EUTRA_NASSteps.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change
function f_EUTRA_TrackingAreaUpdate_Step1_5(EUTRA_CellId_Type               p_CellId,
                                              RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                              EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                              template (present) EstablishmentCause p_EstablishmentCause := ?,
                                              template (omit) PLMN_List p_EPLMNlist := omit,
                                              template B3_Type p_EpsUpdate_TypeValue := omit,
                                              UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,    // @sic R5s120617 sic@
                                              B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                              template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B))
 // @sic R5-165962 sic@ 
    runs on EUTRA_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@
    
    /* Step 1 - 3 */
    /* The TAU Request shall be integrity-protected, but not confidentiality-protected.
       UE uses the cached security context algorithms to protect the TAU Request message.
       ( reuse of keys allocated during attach procedure )
       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */
    /* Check that TRACKING AREA UPDATE REQUEST message is present and conforms
       with default message ( 36.508 Table 4.7.2-27: TRACKING AREA UPDATE REQUEST ) */
    v_NasInd := f_EUTRA_TrackingAreaUpdate_Step1_3(p_CellId, // @sic R5s110632 sic@
                                                   p_RRC_TI,
                                                   p_ForcedAttach,
                                                   p_EstablishmentCause,
                                                   p_EpsUpdate_TypeValue,
                                                   p_UE_NetworkCap, // @sic R5-160777 sic@
                                                   p_ActiveFlag);
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    
    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */
    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,
                                               v_GutiParams,
                                               v_EpsUpdate_TypeValue,
                                               v_EPS_ContextStatus,
                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, // @sic R5s110078 sic@
                                               p_ForcedAttach,
                                               p_EPLMNlist,
                                               -,
                                               p_T3412); // @sic R5-180728 sic@
  }



After change
function f_EUTRA_TrackingAreaUpdate_Step1_5(EUTRA_CellId_Type               p_CellId,
                                              RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,
                                              EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,
                                              template (present) EstablishmentCause p_EstablishmentCause := ?,
                                              template (omit) PLMN_List p_EPLMNlist := omit,
                                              template B3_Type p_EpsUpdate_TypeValue := omit,
                                              UE_NetworkCap_ToBeTested p_UE_NetworkCap := DEFAULT,    // @sic R5s120617 sic@
                                              B1_Type p_ActiveFlag := tsc_EpsUpdate_NotActive,
                                              template (omit) GPRS_Timer p_T3412 := cs_GprsTimer_tv('5A'O, tsc_GprsTimerUnit_deact, '00000'B),
                                              template (value) EPS_NetworkFeatureSupport p_NtwFeatureSupport := cs_EPS_NwkFtSup_IMSVoice) // @sic R5-165962 sic@ //EIEI-Change
    runs on EUTRA_PTC
  {
    var NAS_MSG_Indication_Type v_NasInd;
    var GutiParameters_Type v_GutiParams:= f_EUTRA_CellInfo_GetGuti(p_CellId);
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(p_ForcedAttach); // @sic R5s100029, R5s110176 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus; // @sic R5s100608 sic@
    
    /* Step 1 - 3 */
    /* The TAU Request shall be integrity-protected, but not confidentiality-protected.
       UE uses the cached security context algorithms to protect the TAU Request message.
       ( reuse of keys allocated during attach procedure )
       ( 33.401 cl. 9.1.2 Idle mode procedures in E-UTRAN ) */
    /* Check that TRACKING AREA UPDATE REQUEST message is present and conforms
       with default message ( 36.508 Table 4.7.2-27: TRACKING AREA UPDATE REQUEST ) */
    v_NasInd := f_EUTRA_TrackingAreaUpdate_Step1_3(p_CellId, // @sic R5s110632 sic@
                                                   p_RRC_TI,
                                                   p_ForcedAttach,
                                                   p_EstablishmentCause,
                                                   p_EpsUpdate_TypeValue,
                                                   p_UE_NetworkCap, // @sic R5-160777 sic@
                                                   p_ActiveFlag);
    v_EPS_ContextStatus := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    
    /* Step 4 - 5: send TAU ACCEPT and receive TAU COMPLETE */
    f_EUTRA_TrackingAreaUpdate_Accept_Complete(p_CellId,
                                               v_GutiParams,
                                               v_EpsUpdate_TypeValue,
                                               v_EPS_ContextStatus,
                                               v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType, // @sic R5s110078 sic@
                                               p_ForcedAttach,
                                               p_EPLMNlist,
                                               p_NtwFeatureSupport,//EIEI- Change
                                               p_T3412); // @sic R5-180728 sic@
  }
 

[bookmark: _Toc110976101]2.6 Change 6
	Function name
	Constant tsc_ECallTestService

	Reason for change
	UE will include phone-context parameter in TEL URI built from eCall test number in EFFDNURI

	Summary of change
	Corrected the expected request-URI

	TTCN module
	 IMS_eCall_Functions.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change
  
  const charstring tsc_ECallTestService := "123456"; /* @sic R5-191103 sic@ 36.508 cl. 4.9.5.3
                                                        @status    APPROVED (IMS) */
  //----------------------------------------------------------------------------

  


After change
  
  const charstring tsc_ECallTestService := 123456;phone-context=test.3gpp.com"; /* @sic R5-191103 sic@ 36.508 cl. 4.9.5.3
                                                        @status    APPROVED (IMS) */

[bookmark: _Toc110976102]2.7 Change 7
	Function name
	f_ECall_InitialInvite_AnnexC47()

	Reason for change
	1. For test tpe eCall the call is made within normal registration so the condition A7 (INVITE for creating an emergency session within an emergency registration using IMS security) should not be applied. This will cause expectred route header and others to be set incorrectly.
2. Note 1 in 36.523-1 table Table 11.3.2.3.2-1 indicates that Annex A.2.1 using condition A25 should be used. This should also include accept header to allow eCall data, as per other eCall conditions.

	Summary of change
	1. Apply condition A4 (INVITE creating a dialog) for test type eCall.
2. Extend the check for setting accept header to include test type eCall
Note : An associated prose CR on 34.229-1 will be raised at RAN5#96

	TTCN module
	 IMS_eCall_Functions.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



Before change
---- Skipped code ----

function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC)) { // @sic R5-220971 sic@
      v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
      v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }
} 

After change
---- Skipped code ----
  
function f_ECall_InitialInvite_AnnexC47(ECall_Type p_ECallType) runs on IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var MIME_Message v_MIME_Message;
    var MIME_Part v_MIME_Part;
    var SDP_Message v_SdpOffer;
    var template (present) SDP_Message v_SDP_Message_Step1;
    var template (present) SipUrl v_ECallPSAP_Rx := f_SipUrl_eCall_RX(p_ECallType);
    var template (present) MessageHeader v_MessageHeader;
    var template (present) AcceptBody_List v_AcceptBodyList := { cr_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML), * };    // condition A20, A21
    var template (present) SemicolonParam_List v_DispositionParams := { cr_GenericParam("handling", "optional"), * };             // condition A20, A21
    var boolean v_A8 := pc_IMS_Geolocation;
    var INVITE_A_2_1_Context_Type v_Context := A_2_1_A7;
    if(p_ECallType == TEST)
    {
      v_Context := A_2_1_A4; 
    } 
    var charstring v_CidUrl;
    var ECall_Type v_ECallType := p_ECallType;
    
    // Step 1. Receive INVITE for an eCall
    v_MessageHeader := f_IMS_InviteRequest_MessageHeaderRX(v_ECallPSAP_Rx, -, v_Context, -, v_A8);
    if ((v_ECallType == MANUAL) or (v_ECallType == AUTOMATIC) or (v_ECallType == TEST)) { // @sic R5-220971 sic@
      v_MessageHeader.callInfo := cr_CallInfo(-, "EmergencyCallData.eCall.MSD");     // condition A20, A21 @sic R5-210324: "EmergencyCallData" instead of "emergencyCallData" sic@
      v_MessageHeader.recvInfo := cr_RecvInfo_Def("EmergencyCallData.eCall.MSD");    // condition A20, A21
      v_MessageHeader.accept := cr_Accept(v_AcceptBodyList);                         // condition A20, A21
    }
    v_MessageHeader.contentType := cr_ContentTypeMIME;                             // condition A20, A21, A8, A25 @sic R5-217960 sic@
    v_IMS_DATA_REQ := valueof(f_IMS_INVITE_ReceiveRequest_A_2_1_Optional(v_Context, v_ECallPSAP_Rx, v_MessageHeader));
    v_MIME_Message := v_IMS_DATA_REQ.Request.Invite.messageBody.mimeMessageBody;

    // Check PIDF
    if (v_A8) {
      v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, tsc_MimeType_Pidf);
      f_IMS_DecodeAndCheckGeolocation(v_MIME_Part.body.xmlBody);
    }
    // Check SDP
    v_SDP_Message_Step1 := f_IMS_BuildSDP_EmergencyCallOffer_RX();   // Build RX SDP template using codecs AMR and AMR_WB
    v_MIME_Part := f_IMS_MIME_MessageBody_GetBody(v_MIME_Message, c_sdpApplication);
    v_SdpOffer := f_IMS_SIP_DecodeMatchAndCheckSDP(v_MIME_Part.body.sdpMessageBody, v_SDP_Message_Step1);

    // Check eCall MSD
    v_CidUrl := v_IMS_DATA_REQ.Request.Invite.msgHeader.callInfo.callInfoBody[0].url;
    f_IMS_ECall_MIME_DecodeAndCheckEmergencyCallData(v_MIME_Message, tsc_MimeType_EmergencyCallDataMSD, v_CidUrl, v_DispositionParams);

    return f_IMS_InviteRequestWithSdpMO(v_IMS_DATA_REQ, v_SdpOffer);
  }
} 
[bookmark: _Toc110976103]2.8 Change 8
	Function name
	New function f_IMS_eCallSetup_Send200OK_ReceiveACK_Step12_13

	Reason for change
	200 OK  contents for INVITE should follow condition A13 “Response sent by SS for INVITE for eCall over IMS session with ACK element = TRUE”


	Summary of change
	New function added for 200OK message with control block element with 
“<ack received="true" ref=" addr-spec of the Content-ID of MIME body part containing the MSD sent by the UE in INVITE"/>” as per 34.229-1 A.3.1 with condition A13.

	TTCN module
	 IMS_eCall_Functions.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.




New function

function f_IMS_eCallSetup_Send200OK_ReceiveACK_Step12_13(IMS_InviteRequestWithSdp_Type p_InviteRequestWithSdp) runs on IMS_PTC
  { 
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp := p_InviteRequestWithSdp;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var INVITE_Request v_InviteRequest;
    var template (value) MessageHeader v_MessageHeader_InviteResponse_200;
    var template (value) AcceptBody_List v_AcceptBodyList;
    var template (present) MessageHeader v_MessageHeader_Ack;
    var template (value) SDP_Message v_SDP_Message_Step2;
    var template (value) MessageBody v_MessageBody;
    var IMS_CallType_Type v_EmergencyType := REGULAR_EMERGENCY;
    var charstring v_MSDContentIdRef;
    var charstring v_BoundaryStep2 := "boundary1";
    var charstring v_ContentId := "psap@3gpp.org";
    var charstring v_CidUrl_Step2 := "cid:" & v_ContentId;   /* RFC 8147, RFC 2392 clause 2: cid-url = "cid" ":" content-id
                                                                @sic R5s210849 change 2.1 sic@ */
    var charstring p_ContactUri := px_IMS_CalleeContactUri; //EIEI-Change
    v_InviteRequest := v_InviteRequestWithSdp.Invite;
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    v_MSDContentIdRef := f_IMS_ECall_MIME_GetMSD_ContentId(v_InviteRequest.messageBody);
    
    // Step 12. SS sends 200 OK with SDP contents
    v_MessageHeader_InviteResponse_200 := f_IMS_OtherResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, cs_ContentTypeMIME(v_BoundaryStep2), -, -, -, NORMAL_CALL);
    v_AcceptBodyList := {
      cs_AcceptBody(c_sdpApplication),
      cs_AcceptBody(tsc_MimeType_Pidf),
      cs_AcceptBody(tsc_MimeType_EmergencyCallDataControlXML),
      cs_AcceptBody(tsc_MimeType_EmergencyCallDataMSD)
    };
    v_MessageHeader_InviteResponse_200.accept := cs_Accept(v_AcceptBodyList);
    v_MessageHeader_InviteResponse_200.callInfo := cs_CallInfo(v_CidUrl_Step2, "EmergencyCallData.Control");
    v_MessageHeader_InviteResponse_200.recvInfo := cs_RecvInfo_Def("emergencyCallData.eCall.MSD");
    v_MessageBody := f_IMS_ECall_MIME_MessageBodyTX(v_BoundaryStep2,
                                                    v_ContentId,
                                                    cs_EmergencyCallData_Control_Ack(true, v_MSDContentIdRef));
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader_InviteResponse_200, v_MessageBody)));
    
    // Step 13. UE sends ACK
    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine200, NORMAL_CALL);
    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(f_SIP_BuildSipUri_RX(p_ContactUri), v_MessageHeader_Ack)));
  }  

[bookmark: _Toc110976104]2.9 Change 9
	Function name
	New altstep a_IMS_ReceiveResponse200OK_repeat

	Reason for change
	Following the completiong of IMS registration (receiving 200 OK for REGISTER), the UE may initiate the test eCall immediately, in parallel with subscribing to the reg-event package. In particular the INVITE may be received at the TTCN before the 200 OK response for NOTIFY.


	Summary of change
	New altstep based on a_IMS_ReceiveResponse200OK but including “repeat” statement so that this message can be received and handled while waiting for INVITE or PRACK message during MO call setup.

	TTCN module
	IMS_Procedures_Common.ttcn

	MCC160 Comment
	Postponed. Discussions on prose and TTCN ongoing.



New altstep

altstep a_IMS_ReceiveResponse200OK_repeat(template (value) MessageHeader p_RequestHeader,
                                              template (present) InternetProtocol_Type p_Protocol) runs on IMS_PTC
  { // NOTE: in general the responses e.g. for different NOTIFY messages differ in the CSeq value only
    var template (present) MessageHeader v_MessageHeader_Response := f_IMS_OtherResponse_200_MessageHeaderRX(p_RequestHeader);
    var IMS_DATA_RSP v_IMS_DATA_RSP_ByRef;  // Dummy
    [] a_IMS_ReceiveResponse(p_RequestHeader, p_Protocol, cr_Response(c_statusLine200, v_MessageHeader_Response), v_IMS_DATA_RSP_ByRef)
    {
       repeat;
    }
  }
  







1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc110976105]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on LTE band 4 using LTE ciphering algorithm eea1 and integrity algorithm eia2
3 [bookmark: _Toc110976106]Execution Log Files
3.1 [bookmark: _Toc110976107]Qualcomm SA515M
The Qualcomm SA515M UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
Keysight/TC_11_3_2_LOG.html
· PICS/PIXIT parameter file:
Keysight/TC_11_3_2_PIXIT.xml


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc110976108]References 
	[1]
	R5s220882 :  Supporting information for addition of eCall over IMS test case 11.3.2



